LONDON, METEOROLOGICAL OFFICE.

Met.O.19 Branch Memorandum No.26.

A step-wise multiple regression computer program for the retrieval or stratospheric temperature profiles from satellite radiances. By SLINGO,A.

London,Met.Off.,Met.O.19 Branch Mem. No.26,1976,31cm.Pp.9,pls.17.2 Refs.

An unofficial document - restriction at head of first page to be observed.

Met O 19 Branch Memorandum No.26

A STEP-WISE MULTIPLE REGRESSION COMPUTER PROGRAM FOR THE RETRIEVAL OF STRATOSPHERIC TEMPERATURE PROFILES FROM SATELLITE RADIANCES

A SLINGO

January 1976

1. INTRODUCTION

Many methods of retrieving atmospheric temperature data from satellite radiances have been devised. Three of the more well known methods have been investigated by Goddard and Hunt (1974). One fundamental disadvantage of all these methods is that the weighting functions must be known in order to carry out the retrieval. Once a radiometer is in orbit, however, changes in its specifications are bound to occur and these will be reflected in the weighting functions and, if these changes are not allowed for, could produce systematic errors in the retrievals. This report describes a step-wise multiple regression retrieval program which does not use the weighting functions and which can easily be made independent of long term drifts by updating the regression coefficients with data from coincident radiometer/rocketsonde soundings.

2. PROGRAM DETAILS

A stepwise regression introduces the independent variables one by one, the order being determined by the observed partial correlation with the dependent variable. Independent variables are not allowed to enter the regression if their partial correlation with the dependent variable is too low. This facility is important when the method is used to retrieve profiles from noisy radiances, where the inclusion in the regression of radiances which do not correlate well with the dependent variable would make the retrieval worse.

A stepwise regression is performed between the satellite radiances and the black body radiance of the atmosphere at a given pressure. A temperature profile is constructed by repeating this process at each of the standard pressure levels between 100 and 0.1 mb. There are several reasons why this method is preferred to the direct retrieval of thicknesses;

1. For many applications a temperature profile is preferred to a list of thicknesses; for example in comparisons with radio and rocketsonde data and in checking how small scale features such as the tropopause are retrieved.

2. Any thickness between 100 and 0.1 mb can be found without the regression having to be repeated.

3. All thicknesses calculated from a profile are mutually consistent, which would not be the case if thicknesses were retrieved directly, as each thickness would be retrieved quite independently of any other.

Despite these advantages the program would be useless if it produced much worse retrievals of thickness than a method which carried out a regression to a set of thicknesses directly. The program was therefore re-written to perform such a regression and the results of a comparison with the profile method are presented in section 3.3

The program was tested with the latest collection of rocket and radiosonde temperature profiles first compiled by Barwell and Hoskin (1972). These profiles are in the form of temperatures at the 50 pressure levels defined by;

$$p(n) = 1013.2458\, e^{-\frac{(n-1)}{5}} \qquad n=1,50 \qquad (1)$$

The experiments described in the next section used the 48 profiles in the Winter Pole sample, as of all the seven latitude season samples this shows the greatest variability in the stratosphere and hence should test the retrieval method most severely.

In a preliminary program the radiances which would be observed by a satellite radiometer scanning these temperature profiles were computed using the formulae;

$$\text{RADIANCE IN } i^{th} \text{ CHANNEL} = \sum_{i=1}^{50} W_{iN} B_N \qquad i=1,6 \qquad (2)$$

$$\text{WHERE} \qquad B_N = \frac{c_1 \tilde{\nu}_i^3}{e^{\frac{c_2 \tilde{\nu}_i}{T_n}} - 1} \qquad n=1,50 \qquad (3)$$

Where T_N is the temperature in K at the level whose pressure is p(n), $\tilde{\nu}_i$ is the wavenumber in cm^{-1} of the radiometer channel i, $c_1 = 1.19096 \times 10^{-16}\ Wm^2$, $c_2 = 1.43879 \times 10^{-2}\ m^oK$, and the radiances are measured in radiance units; $1\ RU \equiv 1\ mWm^{-2}sr^{-1}(cm^{-1})$.

The weighting functions, W_{iN}, are those expected for the three SSU and the three highest BSU channels of the Tiros N Operational Vertical Sounder.

It must be stressed that the preliminary program is entirely separate from the regression and was run merely to present to the regression program data which is as similar as possible to that which it is expected to be processing when Tiros N becomes operational.

Print-outs of the profile regression and direct thickness regression programs appear in appendices 1 and 2 respectively. These should be self explanatory, although a few sections will now be discussed in greater detail.

A sample of temperature profiles and radiances is read in and divided into two groups, the first of which is used to compute the regression coefficients which are then used to retrieve all the profiles. The retrieval errors are calculated using the second group only.

There is a facility for simulating the effect of instrumental noise on the retrievals. Noise with zero mean and the standard deviation expected in each channel can be added to the radiances using the subroutines GAUSS and RANDU. From a given integer these subroutines produce the same string of "random" numbers each time they are called, so to simulate real noise as closely as possible this integer was changed at each calling by deriving it from the time of day, recorded by the computer clock in milliseconds, which proved to be sufficiently random.

The temperature profiles are interpolated to the standard pressure levels between 100 and 0.1 mb using a first order interpolation in the logarithm of pressure. The stepwise regression itself is performed by calling the subroutines CORRE, MSTR, LOC and STPRG, which use the simple regression formula;

$$B_j - \bar{B}_j = \sum_{i=1}^{6} a_{ij} \left(R_i - \bar{R}_i \right) \qquad (4)$$

hence;

$$B_j = b_j + \sum_{i=1}^{6} a_{ij} R_i \qquad (5)$$

For details of these subroutines see the Programmer's Manual, IBM Scientific Subroutine Package, version III.

3. EXPERIMENTS

3.1 Number of radiances required in the regression

In order to discover how the retrieval errors depend on the number of variables entered into the regression, the program was run with up to 6 radiances in the regression, the actual radiances used being chosen by the program. In table I the means and standard deviations of the errors of the retrieved heights of the 30, 10, 3, 1, 0.3 and 0.1 mb surfaces with respect to the 100 mb surface are presented. The results are presented in this form because such heights form the basic input data to operational forecasting models.

The results show that if more than four radiances enter the regression there is no significant improvement in the thickness retrieval. For samples with less variability in the temperature profiles, and in the presence of noise, this number is expected to be even lower. To ensure the smallest possible retrieval error with the minimum number of radiances, the program was therefore altered so that at least four radiances enter the regression, the remaining two only being entered if their partial correlation with the dependent variable is high enough.

3.2 The effect of instrumental noise

The program was now run both with and without the noise option described in section 2. Since the total number of profiles in the winter pole sample is only 48, significant variations in the average retrieval errors were expected, depending on which profiles were used in the regression. The experiment was therefore carried out three times, using different subsets of 24 profiles from the winter pole sample in the regression, the numbers of the profiles in each subset being drawn from a table of random integers.

The noise added to the radiances is summarised in Table 2. The results are presented in Table 3, which shows that the thickness retrieval errors vary considerably with the subset chosen. Both with and without noise the heights of the 30, 10, 3 and 1 mb surfaces above 100 mb are retrieved with similar errors,

TABLE I

DEPENDENCE OF THE THICKNESS RETRIEVAL ERRORS ON THE NUMBER OF RADIANCES INCLUDED IN THE REGRESSION, SAMPLE "A" USED

THICKNESS (dm)	NUMBER OF RADIANCES USED IN THE REGRESSION					
	1	2	3	4	5	6
100-0.1	16.2 ± 70.3	-1.6 ± 49.3	-2.4 ± 42.9	0.7 ± 25.8	0.8 ± 27.3	8.2 ± 26.2
100-0.3	17.9 ± 45.1	2.6 ± 18.0	2.0 ± 10.6	-0.7 ± 5.3	-1.3 ± 5.0	-0.5 ± 5.5
100-1	7.5 ± 20.2	4.0 ± 10.0	2.8 ± 8.7	1.6 ± 4.9	-0.2 ± 4.9	-0.8 ± 5.0
100-3	4.6 ± 22.9	3.1 ± 14.2	1.7 ± 5.3	0.0 ± 3.6	-0.5 ± 3.0	0.2 ± 3.1
100-10	7.7 ± 20.9	5.1 ± 11.7	3.6 ± 6.8	1.5 ± 5.0	0.5 ± 4.8	0.4 ± 4.4
100-30	4.5 ± 12.6	2.4 ± 4.5	1.7 ± 3.2	0.9 ± 2.7	-1.0 ± 3.6	-0.8 ± 4.1

TABLE 2

NOISE ADDED TO THE RADIANCES

CHANNEL	SSU			BSU		
	1	2	3	1	2	3
WAVE NUMBERS (CM^{-1})	668.0	668.0	668.0	668.5	678.0	690.0
STANDARD DEVIATION OF NOISE (RU)	1.75	0.7	0.35	0.45	0.16	0.19

TABLE 3

THE EFFECT OF INSTRUMENTAL NOISE ON THE THICKNESS

RETRIEVAL ERRORS FOR THREE SUBSETS OF THE WINTER POLE SAMPLE

	THICKNESS (dm)	A	B	C
WITHOUT NOISE	100-0.1	8.6 ± 26.5	-4.3 ± 35.4	1.5 ± 22.9
	100-0.3	0.1 ± 5.6	1.3 ± 9.1	1.0 ± 6.5
	100-1	0.3 ± 5.5	0.7 ± 2.9	-0.5 ± 4.5
	100-3	0.5 ± 2.6	-0.8 ± 2.9	-1.0 ± 3.0
	100-10	2.0 ± 5.8	2.1 ± 3.0	-0.4 ± 4.4
	100-30	1.0 ± 2.6	-0.8 ± 3.8	0.7 ± 2.2
WITH NOISE ADDED	100-0.1	-1.8 ± 30.0	12.2 ± 45.8	3.6 ± 35.7
	100-0.3	1.2 ± 22.0	2.7 ± 20.9	-0.6 ± 19.5
	100-1	2.1 ± 9.5	-2.6 ± 6.5	-0.7 ± 7.9
	100-3	0.4 ± 4.0	-4.0 ± 4.5	-1.3 ± 6.8
	100-10	0.6 ± 5.9	0.6 ± 4.3	0.3 ± 5.4
	100-30	-0.2 ± 3.4	-1.7 ± 4.8	1.5 ± 3.5

which is probably due to the relatively small variability of the profiles below 1 mb and to the way the retrieved profile oscillates about the actual profile, so that larger thicknesses are not necessarily retrieved with larger errors. Figures 1 and 2 illustrate this effect for two profiles from the subset used to calculate the thickness retrieval errors for sample C.

The deterioration in the accuracy of the retrievals above 1 mb is mainly due to the fact that the peak of the highest weighting function used lies below this level. The radiances therefore contain very little information on what is unfortunately the most variable region of the temperature profile.

Not surprisingly the addition of noise always makes the retrieval worse, though the factor by which the standard deviation of the retrieval error increases is very much a function of the sample used.

3.3 Comparison with direct thickness regression

In the direct regression program a stepwise regression is performed between the required thicknesses and the equivalent temperatures of the measured radiances. As can be seen from appendices 1 and 2, in all other respects the program is identical to the profile regression. The experiment described in section 3.2 was repeated with the direct regression and the results are given in table 4. There are certainly no glaring differences between the numbers in Tables 3 and 4. The differences which do exist tend to show the direct method to be worse than the profile method, though overall the differences are not statistically significant.

FIGURE 1

HEISS ISLAND 29-1-1971 0902 Z



THICKNESSES (dm)

	ACTUAL	RETRIEVED	ERROR
100.0 - 30	754.7	752.1	-2.6
100.0 - 10	1487.4	1486.1	-1.3
100.0 - 3	2310.3	2312.2	1.9
100.0 - 1	3017.7	3029.9	12.2
100.0 - 0.3	3871.3	3847.8	-23.5
100.0 - 0.1	4650.0	4599.0	-51.0

FIGURE 2

HEISS ISLAND 26-2-1971 0930 Z



THICKNESSES (dm)

	ACTUAL	RETRIEVED	ERROR
100.0 - 30	712.0	712.6	0.6
100.0 - 10	1363.6	1356.9	-6.7
100.0 - 3	2119.4	2122.6	3.2
100.0 - 1	2915.4	2937.2	21.8
100.0 - 0.3	3960.6	3951.6	-9.0
100.0 - 0.1	4813.9	4836.9	23.0

TABLE 4

THICKNESS RETRIEVAL ERRORS FOR THE DIRECT THICKNESS REGRESSION PROGRAM

	THICKNESS (dm)	A	B	C
WITHOUT NOISE	100-0.1	8.4 ± 35.9	4.9 ± 31.4	-3.7 ± 25.8
	100-0.3	2.0 ± 12.2	6.8 ± 12.8	-1.4 ± 10.0
	100-1	4.8 ± 8.2	1.4 ± 6.3	-1.0 ± 6.1
	100-3	2.4 ± 5.8	1.9 ± 5.8	-0.4 ± 5.2
	100-10	1.2 ± 6.6	3.8 ± 8.9	-1.2 ± 4.9
	100-30	2.2 ± 3.8	0.9 ± 3.2	-0.2 ± 2.4
WITH NOISE ADDED	100-0.1	-2.1 ± 36.1	1.2 ± 33.2	4.9 ± 42.9
	100-0.3	3.1 ± 21.6	-0.7 ± 16.5	7.9 ± 15.1
	100-1	1.6 ± 9.4	0.0 ± 8.2	-1.4 ± 7.9
	100-3	1.6 ± 8.3	4.0 ± 8.3	-2.8 ± 6.2
	100-10	2.7 ± 8.0	3.4 ± 7.9	-1.2 ± 8.2
	100-30	2.1 ± 3.9	0.6 ± 3.9	0.0 ± 2.9

4. CONCLUSIONS

A computer program which retrieves stratospheric temperature profiles from satellite radiance data using a stepwise multiple regression has been written. The program has been tested with a data set containing a wide range of profile shapes and the effect of instrumental noise on the retrievals has been investigated. Comparison with a version of the program in which thicknesses are regressed directly from the radiance data shows that there is no significant difference between the errors produced by the two methods.

The versatility of the profile regression program makes it a natural choice for an operational retrieval method. In an operational environment the regression coefficients could be up-dated at intervals using co-located rocket/ radiosonde and radiometer data, which would eliminate the effect of the inevitable long term drifts in the radiometer performance.

It has been recognised for some time that, because of the large differences between stratospheric temperature profiles taken from widely separated latitudes, a single set of regression coefficients for the whole world would produce worse results than, say, seven sets each covering a limited latitude band, such as the winter polar sample used in this study. Rather than divide the world into bands it may be better to examine the radiances themselves more closely and use the values of, or patterns in, the radiances to decide which set of coefficients to use. This could lead to improved retrievals at times of stratospheric warmings and would remove the discontinuities which are bound to occur at the edges of latitude bands.

REFERENCES

BARWELL B R and HOSKIN G C (1972). "Vertical temperature profiles of the troposphere and stratosphere on punched cards". Met O 19 Branch memorandum No. 1.

GODDARD J W F and HUNT R D (1974). "Some results on the comparison of retrieval methods - maximum probability, minimum information and Tychonov regularization". Met O 19 Branch memorandum No. 11.

APPENDIX 1

```
C
C
C     QSWMR-----STEPWISE MULTIPLE REGRESSION
C
C
      DIMENSION XBAR(7),STD(7),D(7),B(7),T(7),IDX(7),L(7)
      DIMENSION TITLE(20,48),RAD(6,48),TEMP(50,48),CC(7,50),CHN(6)
      DIMENSION TSL(22,48),TSLEST(22,48),IA(22),SPL(22),G(130),X(336)
      DIMENSION RX(49),R(28),NSTEP(5),ANS(11),ISEL(48),IDXF(7,22),ASE(6)
      DIMENSION RLHT(22),RTHT(22),IHT(42),THICK(21),RTHICK(21),DT(21,48)
      DIMENSION AVDT(21),SDDT(21)
      C1=1.19096E-5
      C2=1.43879
      DATA CHN/668.0,668.0,668.0,668.5,678.0,690.0/
      DATA ASE/1.75,0.7,0.35,0.45,0.16,0.19/
      DATA SPL/100.0,70.0,50.0,40.0,30.0,20.0,15.0,10.0,7.0,5.0,4.0,3.0,
     12.0,1.5,1.0,0.7,0.5,0.4,0.3,0.2,0.15,0.1/
      DATA IA/12,14,16,17,18,20,22,24,25,27,28,30,32,33,35,37,39,40,41,4
     13,45,47/
      DATA IHT/5,1,8,5,12,8,15,12,8,1,12,5,15,8,19,12,12,1,15,5,19,8,22,
     112,15,1,19,5,22,8,19,15,19,1,22,5,22,15,22,1,22,19/
    1 FORMAT(2I5,2I2)
    3 FORMAT (47H1NUMBER OF PROFILES READ IN......................I3)
    9 FORMAT (47H0NUMBER OF PROFILES USED IN THE REGRESSION.....I3)
   10 FORMAT (47H0NUMBER OF VARIABLES...........................I3)
   12 FORMAT('0VARIABLES USED IN THE REGRESSION',/,T10,'0--INDEPENDENT V
     1ARIABLE    1--FORCED INDEPENDENT VARIABLE    2--DELETED VARIABLE
     2  3--DEPENDENT VARIABLE',/,T18,'LEVEL      SSU1     SSU2     SSU3
     3   BSU1      BSU2      BSU3      TEMP',/,22(/T17,F6.2,7I9))
   20 FORMAT(1H0,'****EITHER THE MATRIX IS SINGULAR, OR THE RESIDUAL SUM
     1 OF SQUARES IS NEGATIVE IMPLYING EXTREME ILL CONDITION.',/,' SELEC
     2TION IGNORED.****')
   21 FORMAT(1H0,'****',I6,' OBSERVATIONS ARE TOO FEW TO ALLOW PARAMETER
     1 ESTIMATION FOR',I5,' VARIABLES.  JOB TERMINATED.****')
  203 FORMAT(20A4)
  204 FORMAT(6F6.1)
  220 FORMAT('0TEMPERATURE PROFILES AND RADIANCES(CORRECTED TO 668.0 WAV
     1ENUMBERS). NOTE THAT THE REGRESSION USES PLANCK RADIANCES RATHER T
     2HAN THE',/,' TEMPERATURES THEMSELVES. REGRESSION COEFFICIENTS ARE
     3CALCULATED FROM THE STARRED PROFILES ONLY. THESE ARE THEN USED TO
     4RETRIEVE',/,' ALL THE PROFILES SUBMITTED.'////T18,'TEMPERATURE PRO
     5FILE',T96,'RADIANCES'//T90,'SSU   SSU   SSU   BSU   BSU   BSU'/)
  221 FORMAT(I3,20A4,T87,6F6.1)
  240 FORMAT('1REGRESSION COEFFICIENTS FOR EACH TEMPERATURE LEVEL'//T5,'
     1LEVEL   CONSTANT     SSU',6X,'SSU',6X,'SSU',6X,'BSU',6X,'BSU',6X,'B
     2SU'/)
  241 FORMAT (T4,F6.2,T11,7F9.4)
  402 FORMAT (1H )
      READ (5,83) AST,Z,(G(I),I=1,130)
   83 FORMAT (80A1)
C              READ IN CONTROL DATA
      READ(5,1) N,NC,M,INOISE
      DO 80 I=1,N
   80 ISEL(I)=0
      READ(5,404)((IDXF(I,J),I=1,7),J=1,22)
  404 FORMAT(10(I2,6I1))
```

```
      READ(5,44)(ISEL(I),I=1,NC)
   44 FORMAT(25I3)
      WRITE (6,3) N
      WRITE (6,9) NC
      WRITE (6,10) M
      WRITE(6,12)(SPL(J),(IDXF(I,J),I=1,7),J=1,22)
      IF (NC-M-2) 101,101,102
  101 WRITE (6,21) NC,M
      STOP
  102 MN=M-1
C               READ IN THE PROFILES
  201 DO 202 J=1,N
      READ(5,203)(TITLE(I,J),I=1,20)
      READ(5,204)(RAD(I,J),I=1,MN)
  202 CONTINUE
      READ(11)((TEMP(I,J),I=1,50),J=1,N)
C               CONVERT THE RADIANCES TO 668 WAVENUMBERS
      DO 302 J=1,N
      DO 302 I=1,MN
      EFFT=(C2*CHN(I))/ALOG(1.0+(C1*(CHN(I)**3)/RAD(I,J)))
      RAD(I,J)=3549.99/(EXP(961.112/EFFT)-1.0)
  302 CONTINUE
C               ADD NOISE TO THE RADIANCES IF REQUIRED
      IF(INOISE) 602,602,603
  603 CALL INTVAL(A,0)
      ITIME=INT(A*1000.0)
      IF(ITIME/2.0-ITIME/2-0.3) 604,604,605
  604 ITIME=ITIME+1
  605 WRITE(6,606) ITIME
  606 FORMAT('0NOISE IS ADDED TO THE RADIANCES USING SUBROUTINE GAUSS. T
     1HE SEED IS DERIVED FROM THE TIME AT WHICH THE PROGRAM IS RUN.',/,'
     2 SEED USED=',I9)
      DO 600 J=1,N
      DO 601 I=1,MN
      SD=ASE(I)
      CALL GAUSS(ITIME,SD,0.0,V)
  601 RAD(I,J)=RAD(I,J)+V
  600 CONTINUE
  602 WRITE(6,220)
      LL=1
      DO 40 J=1,N
      WRITE (6,221)(J,(TITLE(I,J),I=1,20),(RAD(K,J),K=1,MN))
      IF (J.EQ.ISEL(LL)) GO TO 41
      GO TO 40
   41 WRITE (6,42) AST
   42 FORMAT ('+',T86,A1)
      LL=LL+1
   40 CONTINUE
C               INITIALISE THE REGRESSION COEFFICIENT MATRIX
      DO 300 IK=1,50
      DO 300 I=1,M
      CC(I,IK)=0.0
  300 CONTINUE
C               FOR PROFILES IN REGRESSIOM PUT RADIANCES INTO X ARRAY
      IL=0
      DO 207 K=1,MN
```

```
      DO 207 J=1,NC
      IL=IL+1
      I=ISEL(J)
      X(IL)=RAD(K,I)
  207 CONTINUE
C              START REGRESSION AT STANDARD PRESSURE LEVELS
      IK=0
C              INTERPOLATE ALL THE PROFILES TO CURRENT STANDARD LEVEL
  206 IK=IK+1
      IB=IA(IK)
      COEF1=5.0*ALOG(SPL(IK)*EXP(0.2*FLOAT(IB))/1013.2458)
      COEF2=1.0-COEF1
      DO 209 J=1,N
  209 TSL(IK,J)=COEF1*TEMP(IB,J)+COEF2*TEMP(IB+1,J)
C              CHANGE TEMPERATURES TO PLANCK RADIANCES
      IL=NC*MN
      DO 208 J=1,NC
      IL=IL+1
      I=ISEL(J)
  208 X(IL)=3549.99/(EXP(961.112/TSL(IK,I))-1.0)
C              PERFORM REGRESSION
      CALL CORRE (NC,M,1,X,XBAR,STD,RX,R,B,D,T)
  140 CALL MSTR(RX,R,M,0,1)
      DO 156 J=1,M
  156 IDX(J)=IDXF(J,IK)
  157 CALL STPRG (M,NC,RX,XBAR,IDX,0.0,NSTEP,ANS,L,B,STD,T,D,IER)
      IF (IER) 158,159,158
  158 WRITE (6,20)
      GO TO 206
  159 CONTINUE
C              ENTER RESULTS INTO REGRESSION COEFFICIENT MATRIX
      CC(1,IK)=ANS(9)
      K=NSTEP(4)
      DO 231 J=1,K
      KK=L(J)
  231 CC(KK+1,IK)=B(J)
      IF(IK-22) 206,250,250
C              WRITE OUT THE REGRESSION COEFFICIENTS
  250 WRITE(6,240)
      WRITE(6,241)(SPL(I),(CC(J,I),J=1,M),I=1,22)
C              WRITE OUT THE RESULTS FOR EACH PROFILE
      DO 500 J=1,N
      WRITE (6,400)(TITLE(I,J),I=1,20)
  400 FORMAT(1H1,20A4)
      WRITE (6,401)
  401 FORMAT('0','PRESSURE',T22,'TEMPERATURES'/,'     MB',T14,'ACTUAL
     1RETRIEVED     DIFF',T70,'0=ACTUAL    *=RETRIEVED'////)
      DO 430 K=1,22
      KS=23-K
      IF(KS.EQ.22) GO TO 417
      LINES=INT(0.5+13.0*ALOG10(SPL(KS)/SPL(KS+1)))
      DO 413 I=1,LINES
      WRITE (6,402)
  413 CONTINUE
  417 CONTINUE
      TSLEST(KS,J)=CC(1,KS)
```

```
      DO 403 I=2,M
      TSLEST(KS,J)=TSLEST(KS,J)+RAD(I-1,J)*CC(I,KS)
  403 CONTINUE
      TSLEST(KS,J)=961.112/(ALOG(1.0+3549.99/TSLEST(KS,J)))
  410 DIFF=TSLEST(KS,J)-TSL(KS,J)
      M1=INT((TSL(KS,J)-159.0)/2)+40
      N1=INT((TSLEST(KS,J)-159.0)/2)+40
      WRITE (6,414) SPL(KS),TSL(KS,J),TSLEST(KS,J),DIFF,SPL(KS),SPL(KS)
  414 FORMAT('+',T2,F6.2,T14,F6.2,T25,F6.2,T35,F6.2,T44,F6.2,T124,F6.2)
      WRITE (6,415)(G(I),I=1,M1),Z
      WRITE (6,415)(G(I),I=1,N1),AST
  415 FORMAT ('+',130A1)
  430 CONTINUE
      WRITE (6,416)
  416 FORMAT(49X,'--|---------|---------|---------|---------|---------|-
     1---------|---------|-'/,' ',T51,'180',T61,'200',T71,'220',T81,'240'
     2,T91,'260',T101,'280',T111,'300',T121,'320 DEG K'/)
C             CALCULATE HEIGHTS ABOVE 100MB
      DO 50 I=1,22
      IF (I-2) 51,52,52
   51 RLHT(I)=0.0
      RTHT(I)=0.0
      GO TO 53
   52 RLHT(I)=RLHT(I-1)+1.46448*(ALOG(SPL(I-1)/SPL(I)))*(TSL(I,J)+TSL(I-
     11,J))
      RTHT(I)=RTHT(I-1)+1.46448*(ALOG(SPL(I-1)/SPL(I)))*(TSLEST(I,J)+TSL
     1EST(I-1,J))
   53 CONTINUE
   50 CONTINUE
C             CALCULATE THE REQUIRED THICKNESSES
      K=0
      DO 54 I=1,21
      K=K+1
      IK=IHT(K)
      K=K+1
      IL=IHT(K)
      THICK(I)=RLHT(IK)-RLHT(IL)
      RTHICK(I)=RTHT(IK)-RTHT(IL)
      DT(I,J)=RTHICK(I)-THICK(I)
   54 CONTINUE
      WRITE (6,55)
   55 FORMAT(' PRESSURE',7X,'THICKNESSES',6X,'PRESSURE',7X,'THICKNESSES'
     1,6X,'PRESSURE',7X,'THICKNESSES',6X,'PRESSURE',7X,'THICKNESSES',/,'
     2 INTERVAL  ACTUAL  RETR''D  DIFF. INTERVAL  ACTUAL  RETR''D  DIFF.
     3 INTERVAL  ACTUAL  RETR''D  DIFF. INTERVAL  ACTUAL  RETR''D  DIFF.
     4'/)
      WRITE(6,56)(THICK(I),RTHICK(I),DT(I,J),I=1,4)
   56 FORMAT(' 100-30 ',3F8.1,'  30-10 ',3F8.1,'  10-3  ',3F8.1,'   3-1
     1 ',3F8.1)
      WRITE(6,57)(THICK(I),RTHICK(I),DT(I,J),I=5,8)
   57 FORMAT(' 100-10 ',3F8.1,'  30-3  ',3F8.1,'  10-1  ',3F8.1,'   3-0.
     13',3F8.1)
      WRITE(6,58)(THICK(I),RTHICK(I),DT(I,J),I=9,12)
   58 FORMAT(' 100-3  ',3F8.1,'  30-1  ',3F8.1,'  10-0.3',3F8.1,'   3-0.
     11',3F8.1)
      WRITE(6,59)(THICK(I),RTHICK(I),DT(I,J),I=13,16)
```

```
 59 FORMAT(' 100-1   ',3F8.1,'  30-0.3',3F8.1,'  10-0.1',3F8.1,'   1-0.
   13',3F8.1)
    WRITE(6,60)(THICK(I),RTHICK(I),DT(I,J),I=17,19)
 60 FORMAT(' 100-0.3',3F8.1,'  30-0.1',3F8.1,T97,'   1-0.1',3F8.1)
    WRITE(6,61)(THICK(I),RTHICK(I),DT(I,J),I=20,21)
 61 FORMAT(' 100-0.1',3F8.1,T97,' 0.3-0.1',3F8.1)
500 CONTINUE
C            CALCULATE THE THICKNESS ERRORS FOR THE PROFILES NOT
C            USED IN THE REGRESSION
    DO 700 I=1,21
    AVDT(I)=0.0
    SDDT(I)=0.0
    LL=1
    DO 701 J=1,N
    IF(J-ISEL(LL)) 702,703,702
703 LL=LL+1
    GO TO 701
702 AVDT(I)=AVDT(I)+DT(I,J)
701 CONTINUE
    AVDT(I)=AVDT(I)/(N-NC)
    LL=1
    DO 704 J=1,N
    IF(J-ISEL(LL)) 705,706,705
706 LL=LL+1
    GO TO 704
705 SDDT(I)=SDDT(I)+(DT(I,J)-AVDT(I))*(DT(I,J)-AVDT(I))/(N-NC-1)
704 CONTINUE
    SDDT(I)=SQRT(SDDT(I))
700 CONTINUE
    IP=N-NC
    WRITE (6,707) IP
707 FORMAT('1STATISTICS OF THE THICKNESS ERRORS FOR THE ',I2,' PROFILE
   1S NOT USED IN THE REGRESSION.'///T8,' PRESSURE     THICKNESS ERROR
   2S'/T8,' INTERVAL       MEAN        S.D.'/)
    DO 710 I=1,21
    J=2*I-1
    K=IHT(J)
    M=IHT(J+1)
    WRITE(6,711) SPL(M),SPL(K),AVDT(I),SDDT(I)
711 FORMAT(T8,F5.1,'-',F4.1,T21,F7.3,T32,F7.3)
710 CONTINUE
    STOP
    END
```

```
      SUBROUTINE DATA
      RETURN
      END

      SUBROUTINE STOUT (NSTEP,ANS,L,B,S,T,NSTOP)
      DIMENSION NSTEP(1),ANS(1),L(1),B(1),S(1),T(1)
      IF(NSTEP(4)-NSTEP(2)) 50,51,51
   50 GO TO 55
   51 IF(NSTEP(4)-4) 52,53,53
   52 GO TO 55
   53 IF(ANS(2)-0.010) 54,55,55
   54 NSTOP=1
      GO TO 43
   55 NSTOP=0
   43 RETURN
      END

      SUBROUTINE GAUSS(IX,S,AM,V)
      A=0.0
      DO 50 I=1,12
      CALL RANDU(IX,IY,Y)
      IX=IY
   50 A=A+Y
      V=(A-6.0)*S+AM
      RETURN
      END

      SUBROUTINE RANDU(IX,IY,YFL)
      IY=IX*65539
      IF(IY)5,6,6
    5 IY=IY+2147483647+1
    6 YFL=IY
      YFL=YFL*.4656613E-9
      RETURN
      END




      INCLUDE OBJMOD(SSPCORR,SSPMSTR,SSPSTPR,SSPLOC)
      INCLUDE PROGLIB(INTVAL)
```

APPENDIX 2

```
C
C
C     QRTHK-----DIRECT THICKNESS REGRESSION
C
C
      DIMENSION XBAR(7),STD(7),D(7),B(7),T(7),IDX(7),L(7),TSLEST(21,48)
      DIMENSION TITLE(20,48),RAD(6,48),TEMP(50,48),CC(7,50),CHN(6)
      DIMENSION TSL(22,48),IA(22),SPL(22),G(130),X(336)
      DIMENSION RX(49),R(25),NSTEP(5),ANS(11),ISEL(48),IDXF(7,22),ASE(6)
      DIMENSION RLHT(22),IHT(42),DT(21,48)
      DIMENSION AVDT(21),SDDT(21)
      C1=1.19096E-5
      C2=1.43879
      DATA CHN/668.0,668.0,668.0,668.5,678.0,690.0/
      DATA ASE/1.75,0.7,0.35,0.45,0.16,0.19/
      DATA SPL/100.0,70.0,50.0,40.0,30.0,20.0,15.0,10.0,7.0,5.0,4.0,3.0,
     12.0,1.5,1.0,0.7,0.5,0.4,0.3,0.2,0.15,0.1/
      DATA IA/12,14,16,17,18,20,22,24,25,27,28,30,32,33,35,37,39,40,41,4
     13,45,47/
      DATA IHT/5,1,8,1,12,1,15,1,19,1,22,1,8,5,12,5,15,5,19,5,22,5,12,8,
     115,8,19,8,22,8,15,12,19,12,22,12,19,15,22,15,22,19/
    1 FORMAT(2I5,2I2)
    3 FORMAT (47H1NUMBER OF PROFILES READ IN....................I3)
    9 FORMAT (47H0NUMBER OF PROFILES USED IN THE REGRESSION.....I3)
   10 FORMAT (47H0NUMBER OF VARIABLES..........................I3)
   12 FORMAT('0VARIABLES USED IN THE REGRESSION',/,T10,'0--INDEPENDENT V
     1ARIABLE     1--FORCED INDEPENDENT VARIABLE     2--DELETED VARIABLE
     2  3--DEPENDENT VARIABLE',/,T10,'PRESSURE INTERVAL    SSU1      SSU2
     3      SSU3      BSU1      BSU2      BSU3    THICKNESS'/)
   20 FORMAT(1H0,'****EITHER THE MATRIX IS SINGULAR, OR THE RESIDUAL SUM
     1 OF SQUARES IS NEGATIVE IMPLYING EXTREME ILL CONDITION.',/,' SELEC
     2TION IGNORED.****')
   21 FORMAT(1H0,'****',I6,' OBSERVATIONS ARE TOO FEW TO ALLOW PARAMETER
     1 ESTIMATION FOR',I5,' VARIABLES.  JOB TERMINATED.****')
  203 FORMAT(20A4)
  204 FORMAT(6F6.1)
  220 FORMAT('0TEMPERATURE PROFILES AND RADIANCES. IN THIS PROGRAM THE T
     1HICKNESSES BETWEEN THE STANDARD LEVELS ARE REGRESSED AGAINST THE R
     2ADIANCE',/,' EQUIVALENT TEMPERATURES. REGRESSION COEFFICIENTS ARE
     3CALCULATED FROM THE STARRED PROFILES ONLY. THESE ARE THEN USED TO
     4RETRIEVE',/,' ALL THE PROFILES SUBMITTED.'////T18,'TEMPERATURE PRO
     5FILE',T96,'RADIANCES'//T90,'SSU    SSU    SSU    BSU    BSU    BSU'/)
  221 FORMAT(I3,20A4,T87,6F6.1)
  240 FORMAT('1REGRESSION COEFFICIENTS FOR EACH THICKNESS'//T5,'THICKNES
     1S',T24,'CONSTANT     SSU',6X,'SSU',6X,'SSU',6X,'BSU',6X,'BSU',6X,'B
     2SU'/)
      READ (5,83) AST,Z,(G(I),I=1,130)
   83 FORMAT (80A1)
C             READ IN CONTROL DATA
      READ (5,1) N,NC,M,INOISE
      DO 80 I=1,N
   80 ISEL(I)=0
      READ(5,404)((IDXF(I,J),I=1,7),J=1,21)
  404 FORMAT(10(I2,6I1))
      READ(5,44)(ISEL(I),I=1,NC)
   44 FORMAT(25I3)
```

```
      WRITE (6,3) N
      WRITE (6,9) NC
      WRITE (6,10) M
      WRITE (6,12)
      DO 800 I=1,21
      J=2*I-1
      K=IHT(J)
      MM=IHT(J+1)
      WRITE (6,11)(SPL(MM),SPL(K),(IDXF(IJ,I),IJ=1,M))
   11 FORMAT(T13,F5.1,'-',F4.1,7I9)
  800 CONTINUE
      IF (NC-M-2) 101,101,102
  101 WRITE (6,21) NC,M
      STOP
  102 MN=M-1
C               READ IN THE PROFILES
  201 DO 202 J=1,N
      READ(5,203)(TITLE(I,J),I=1,20)
      READ(5,204)(RAD(I,J),I=1,MN)
  202 CONTINUE
      READ(11)((TEMP(I,J),I=1,50),J=1,N)
C               ADD NOISE TO THE RADIANCES IF REQUIRED
      IF(INOISE) 602,602,603
  603 CALL INTVAL(A,0)
      ITIME=INT(A*1000.0)
      IF(ITIME/2.0-ITIME/2-0.3) 604,604,605
  604 ITIME=ITIME+1
  605 WRITE(6,606) ITIME
  606 FORMAT('ONOISE IS ADDED TO THE RADIANCES USING SUBROUTINE GAUSS. T
     1HE SEED IS DERIVED FROM THE TIME AT WHICH THE PROGRAM IS RUN.',/,'
     2 SEED USED=',I9)
      DO 600 J=1,N
      DO 601 I=1,MN
      SD=ASE(I)
      CALL GAUSS(ITIME,SD,0.0,V)
  601 RAD(I,J)=RAD(I,J)+V
  600 CONTINUE
  602 WRITE(6,220)
      LL=1
      DO 40 J=1,N
      WRITE (6,221)(J,(TITLE(I,J),I=1,20),(RAD(K,J),K=1,MN))
      IF (J.EQ.ISEL(LL)) GO TO 41
      GO TO 40
   41 WRITE (6,42) AST
   42 FORMAT ('+',T86,A1)
      LL=LL+1
   40 CONTINUE
C               CONVERT THE RADIANCES TO EQUIVALENT TEMPERATURES
      DO 302 J=1,N
      DO 302 I=1,MN
      RAD(I,J)=(C2*CHN(I))/ALOG(1.0+(C1*(CHN(I)**3)/RAD(I,J)))
  302 CONTINUE
C               INITIALISE REGRESSION COEFFICIENT MATRIX
      DO 300 IK=1,21
      DO 300 I=1,M
      CC(I,IK)=0.0
```

```
  300 CONTINUE
C             CALCULATE THE ACTUAL THICKNESSES OF ALL THE PROFILES
      DO 801 J=1,N
C             INTERPOLATE TO STANDARD LEVELS(LOGARITHMIC)
      DO 209 IK=1,22
      IB=IA(IK)
      COEF1=5.0*ALOG(SPL(IK)*EXP(0.2*FLOAT(IB))/1013.2458)
      COEF2=1.0-COEF1
  209 TSL(IK,J)=COEF1*TEMP(IB,J)+COEF2*TEMP(IB+1,J)
C             CALCULATE HEIGHTS ABOVE 100MB
      DO 50 I=1,22
      IF (I-2) 51,52,52
   51 RLHT(I)=0.0
      GO TO 53
   52 RLHT(I)=RLHT(I-1)+1.46448*(ALOG(SPL(I-1)/SPL(I)))*(TSL(I,J)+TSL(I-
     11,J))
   53 CONTINUE
   50 CONTINUE
C             CALCULATE THE REQUIRED THICKNESSES
      K=0
      DO 54 I=1,21
      K=K+1
      IK=IHT(K)
      K=K+1
      IL=IHT(K)
      TSL(I,J)=RLHT(IK)-RLHT(IL)
   54 CONTINUE
  801 CONTINUE
C             FOR THE PROFILES NOMINATED TO ENTER THE REGRESSION
C             PUT RADIANCE EQUIVALENT TEMPLRATURES INTO X ARRAY
      IL=0
      DO 207 K=1,MN
      DO 207 J=1,NC
      IL=IL+1
      I=ISEL(J)
      X(IL)=RAD(K,I)
  207 CONTINUE
C             PERFORM REGRESSION FOR EACH THICKNESS
      DO 803 IK=1,21
  206 IL=NC*MN
      DO 208 J=1,NC
      IL=IL+1
      I=ISEL(J)
  208 X(IL)=TSL(IK,I)
      CALL CORRE (NC,M,1,X,XBAR,STD,RX,R,B,D,T)
  140 CALL MSTR(RX,R,M,0,1)
      DO 156 J=1,M
  156 IDX(J)=IDXF(J,IK)
  157 CALL STPRG (M,NC,RX,XBAR,IDX,0.0,NSTEP,ANS,L,B,STD,T,D,IER)
      IF (IER) 158,159,158
  158 WRITE (6,20)
      IK=IK+1
      GO TO 206
  159 CONTINUE
C             ENTER RESULTS INTO REGRESSION COEFFICIENT MATRIX
      CC(1,IK)=ANS(9)
```

```
      K=NSTEP(4)
      DO 231 J=1,K
      KK=L(J)
  231 CC(KK+1,IK)=B(J)
  803 CONTINUE
C               WRITE OUT THE REGRESSION COEFFICIENTS
      WRITE (6,240)
      DO 805 I=1,21
      J=2*I-1
      K=IHT(J)
      MM=IHT(J+1)
      WRITE (6,241)(SPL(MM),SPL(K),(CC(IJ,I),IJ=1,M))
  241 FORMAT (T5,F5.1,'-',F4.1,T22,7F9.4)
  805 CONTINUE
C               WRITE OUT THE RESULTS FOR EACH PROFILE
      LL=1
      DO 500 J=1,N
      IF(J/2.0-J/2-0.3) 806,806,807
  807 WRITE (6,808)
  808 FORMAT (1H1)
  806 WRITE (6,400)(TITLE(I,J),I=1,20)
  400 FORMAT(1H0,20A4)
      WRITE (6,811)
  811 FORMAT(1H ,T8,'PRESSURE',T34,'THICKNESSES',/,T8,'INTERVAL',T25,'AC
     1TUAL',T37,'RETRIEVED',T52,'DIFF.',/)
      IF (J.EQ.ISEL(LL)) GO TO 809
      GO TO 810
  809 WRITE (6,42) AST
      LL=LL+1
  810 CONTINUE
C               PRINT OUT THE TABLE OF THICKNESSES
      DO 430 K=1,21
      TSLEST(K,J)=CC(1,K)
      DO 403 I=2,M
      TSLEST(K,J)=TSLEST(K,J)+RAD(I-1,J)*CC(I,K)
  403 CONTINUE
      DT(K,J)=TSLEST(K,J)-TSL(K,J)
      I=2*K-1
      IK=IHT(I)
      IM=IHT(I+1)
      WRITE (6,414) SPL(IM),SPL(IK),TSL(K,J),TSLEST(K,J),DT(K,J)
  414 FORMAT (T7,F5.1,'-',F4.1,T18,3F13.1)
  430 CONTINUE
  500 CONTINUE
C               CALCULATE THE THICKNESS RETRIEVAL ERRORS FOR THE PROFILES
C               NOT USED IN THE REGRESSION
      DO 700 I=1,21
      AVDT(I)=0.0
      SDDT(I)=0.0
      LL=1
      DO 701 J=1,N
      IF(J-ISEL(LL)) 702,703,702
  703 LL=LL+1
      GO TO 701
  702 AVDT(I)=AVDT(I)+DT(I,J)
  701 CONTINUE
```

```
      AVDT(I)=AVDT(I)/(N-NC)
      LL=1
      DO 704 J=1,N
      IF(J-ISEL(LL)) 705,706,705
  706 LL=LL+1
      GO TO 704
  705 SDDT(I)=SDDT(I)+(DT(I,J)-AVDT(I))*(DT(I,J)-AVDT(I))/(N-NC-1)
  704 CONTINUE
      SDDT(I)=SQRT(SDDT(I))
  700 CONTINUE
      IP=N-NC
      WRITE (6,707) IP
  707 FORMAT('1STATISTICS OF THE THICKNESS ERRORS FOR THE ',I2,' PROFILE
     1S NOT USED IN THE REGRESSION.'///T8,' PRESSURE      THICKNESS ERROR
     2S'/T8,' INTERVAL        MEAN        S.D.'/)
      DO 710 I=1,21
      J=2*I-1
      K=IHT(J)
      M=IHT(J+1)
      WRITE(6,711) SPL(M),SPL(K),AVDT(I),SDDT(I)
  711 FORMAT(T8,F5.1,'-',F4.1,T21,F7.3,T32,F7.3)
  710 CONTINUE
      STOP
      END
```